

URI986-23

1986/23. GRVDOC - A Fortran program to extract and plot survey locations by identifier

J.W. Hudspeth

*Abstract*

For each contiguous group of records having a common identifier within a database file, Fortran program GRVDOC creates a plotfile showing station locations.

INTRODUCTION

This program was written to provide documentation on the gravity data base file of Tasmania. For each contiguous group of records having a common identifier, the program will create a station location plotfile. All plotfiles are created in the one run of the program and the only operator intervention required is to start the run and to have modified the input file if necessary. The program is written in PE Fortran 77 and runs on the Tasmania Department of Mines Perkin Elmer 3200 computer. The program may easily be adapted to produce location plotfiles by survey for any other similar data base file.

GRVDOC

To commence a GRVDOC run, type "GRVDOC" to start a .CSS of that name. No other operator action is required. The eight character identifier in the first record is read and the reference identifier is set equal to it. Successive records are then read and each record's identifier is compared with the reference. If the identifier is the same as the reference, then the X and Y co-ordinates are stored in arrays and the current maximum and minimum of the group updated if necessary. If the identifier is different from the reference identifier, then the group is closed and the identifier becomes the new reference identifier for the next group. At this stage the maxima and minima for the group are known and the X and Y co-ordinates of each group record are stored in arrays. Based on the range of X and Y co-ordinates, the plot is now scaled to fit both a predetermined grid on A4 paper (see fig. 1) and also one of the allowable (built in) scales. The available scales are 1 to 5000, 10 000, 25 000, 50 000, 100 000, 250 000, 500 000, 1 000 000, 2 500 000, or 5 000 000. The actual area of the predetermined grid occupied by points is always as close as possible (to the nearest predetermined grid square) towards the bottom left hand (SW) corner.

After the first plotfile has been created, the program continues reading the file from the first record of the new group (the record which caused the change of reference identifier) and reads records until a change of identifier again occurs, at which point the second plotfile is created in exactly the same way as already described. The process of creating plotfiles from contiguous records having a common identifier is continued until the end of the input file is reached.

The program needs to keep track of its progress through the input file so that it may multiple read (as is required) the first record of a group. This requires random access reads. However, unless the record is the first of the group, it is read by sequential read so that the program may be properly terminated on reaching the end of the input file without needing to know how many records are contained in the input file.

Identifiers in this input file (TASTOT.BA) have been filled to eight characters with asterisks. Leading asterisks are removed and other

asterisks or illegal characters replaced by "X"s for the purposes of creating a legal plotfile name. It does not matter if the input file has eight character identifier fillers other than asterisks because the program will still replace any illegal character by an "X". The plot title appearing on the A4 sheet is the identifier verbatim as per the record, including fillers.

In order that the program may run efficiently, it reads through the input file once only. It is therefore most important that records within the input file having a common identifier form a contiguous block and are not scattered throughout the file. If there is more than one group of records with the same common identifier then each time such a group is read a plotfile will be created which will overwrite the plotfile of the same name which would have been created when the previous such group was read.

If it is desired to run this program with records having a different format to that expected, it is only necessary to alter the appropriate read format statements accordingly. However, it will be necessary to keep eight characters for the identifier because this is the number expected for the plotfile name. It would be possible to alter the read format statement to read only eight characters of a longer identifier but it should be pointed out that there is a possibility that identifier truncation in this manner might lead to different identifiers becoming the same after truncation; the second group read would then cause obliteration of the plotfile created for the first group.

The plotfile title contains the station identifier, the plot scale and the number of station locations. This information is also listed on the printer.

[2 April 1986]

APPENDIX 1  
GRVDOC listing

```

C*****
C   A PROGRAM TO PLOT LOCATION MAPS FROM THE GRAVITY DATA BASE
C   BY SURVEY.
C
REAL NORTH(4000),NMAX,NMIN,EAST(4000),EMAX,EMIN
INTEGER DUM
DIMENSION IL(4),IS(6),ISX(3),ISY(4)
LOGICAL END
DATA ISX/4HMETR,4HES ,4HEAST/
DATA ISY/4HMETR,4HES ,4HNORT,4HH /
DATA IL/4HAAA,4HAAA,4H ,4H 1:/
DATA IS/4HSTAT,4HION ,4HLOCA,4HTION,4HS, A,4HMG /
CHARACTER*8 TITLE, TITOLD
CHARACTER*12 FNAME

C   ***
C   INITIALISE
C   ***
      END=.FALSE.
      KOUNT=1
C*****
C READ FIRST RECORD OF GROUP, DETERMINE PLOTFILE NAME, SET
C   UP STARTING VALUES FOR REPEATED READ OF GROUP RECORDS
C*****
      20 N=1
         READ(2,30,REC=KOUNT) EAST(1),NORTH(1),TITLE
         READ(2,29,REC=KOUNT) IL(1),IL(2)
      29 FORMAT(72X,2A4)
      30 FORMAT(10X,2F9.1,44X,A8)
         TITOLD=TITLE

C   *****
C   REMOVE ASTERISKS FROM FRONT OF IDENTIFIER
C   *****
      DO 80 J=1,8
         IF(TITLE(J:J).NE.'*') THEN
            DUM=J
            GO TO 90
         ENDIF
         IF(J.EQ.8) THEN
            WRITE(5,75)
      75 FORMAT(1X,'TITLE ALL ASTERISKS')
            GO TO 900
         ENDIF
      80 CONTINUE

C   *****
C   REPLACE ILLEGAL FILENAME CHARACTERS IN IDENTIFIER BY 'X'
C   *****
      90 IVAL=ICHAR(TITLE(J:J))
         IF(IVAL.LT.65.OR.IVAL.GT.90) TITLE(DUM:DUM)='X'
         IF(DUM.LT.8) THEN
            DO 96 J=DUM+1,8
               IVAL=ICHAR(TITLE(J:J))
               IF(IVAL.LT.48.OR.(IVAL.LT.65.AND.IVAL.GT.57).OR.IVAL.GT.90) THEN
                  TITLE(J:J)='X'
               ENDIF
            96 CONTINUE
            ENDIF
            FNAME=TITLE(DUM:8)//'.GTM'
            EMIN=EAST(1)
            EMAX=EAST(1)
            NMIN=NORTH(1)
            NMAX=NORTH(1)

```

```

C*****
C REPEATED READ OF GROUP RECORDS - UPDATE MAX & MIN AS REQUIRED
C*****
120 N=N+1
READ(2,30,END=600) EAST(N),NORTH(N), TITLE
IF(TITLE.NE.TITOLD) GO TO 200
KOUNT=KOUNT+1
IF(NORTH(N).GT.NMAX) NMAX=NORTH(N)
IF(NORTH(N).LT.NMIN) NMIN=NORTH(N)
IF(EAST(N).GT.EMAX) EMAX=EAST(N)
IF(EAST(N).LT.EMIN) EMIN=EAST(N)
GO TO 120
C*****
C END GROUP. CALCULATE SCALE
C*****
200 N=N-1.
YINT=(NMAX-NMIN)
XINT=(EMAX-EMIN)
IF(XINT/YINT.LT.0.7) THEN
TINT=YINT
BOUND=9.0
ELSE
TINT=XINT
BOUND=6.0
ENDIF
IF(TINT.LT.BOUND*1000) THEN
IF(TINT.LT.BOUND*100) THEN
C SCALE 1:5000 *****
ISCALE=5000
ELSE
IF(TINT.LT.BOUND*200) THEN
C SCALE 1:10000 *****
ISCALE=10000
ELSE
IF(TINT.LT.BOUND*500) THEN
C SCALE 1:25000 *****
ISCALE=25000
ELSE
C SCALE 1:50000 *****
ISCALE=50000
ENDIF
ENDIF
ELSE
IF(TINT.GE.BOUND*10000) THEN
IF(TINT.LT.BOUND*20000) THEN
C SCALE 1:1000000 *****
ISCALE=1000000
ELSE
IF(TINT.LT.BOUND*50000) THEN
C SCALE 1:2500000 *****
ISCALE=2500000
ELSE
C SCALE 1:5000000 *****
ISCALE=5000000
ENDIF
ENDIF
ELSE
IF(TINT.LT.BOUND*2000) THEN
C SCALE 1:100000 *****
ISCALE=100000
ELSE
IF(TINT.LT.BOUND*5000) THEN
C SCALE 1:250000 *****
ISCALE=250000
ELSE

```

```


C      SCALE=500000          *****
      ISCALE=500000
      ENDIF
      ENDIF
      ENDIF
      ENDIF
      INC=ISCALE/50
C*****
C  CREATE PLOT FILE
C*****
      XMIN=INC*INT(EMIN/INC)
      YMIN=INC*INT(NMIN/INC)
      FACTOR=39.37/ISCALE
      CALL PLTOUT(FNAME)
      CALL INITAL(9,200,36,0,0,0)
      XOFF=0.5
      YOFF=2.25
      X1=XOFF
      X2=XOFF+14.0/2.54
      DO 240 J=1,11
      Y=YOFF+(J-1)*2.0/2.54
      CALL PLOT(X1,Y,3)
      CALL PLOT(X2,Y,2)
240  CONTINUE
      Y1=YOFF
      Y2=YOFF+20.0/2.54
      DO 260 J=1,8
      X=XOFF+(J-1)*2.0/2.54
      CALL PLOT(X,Y1,3)
      CALL PLOT(X,Y2,2)
260  CONTINUE
      DO 270 K=1,8
      X=XOFF+(K-1)*2.0/2.54
      Y=YOFF-0.1
      INUM=XMIN+INC*(K-1)
      FNUM=1.0*INUM
      CALL NUMBER(X,Y,.07,FNUM,270.0,-1)
270  CONTINUE
      DO 280 K=1,11
      Y=YOFF+(K-1)*2.0/2.54
      X=XOFF+0.1+7*2.0/2.54
      INUM=YMIN+INC*(K-1)
      FNUM=1.0*INUM
      CALL NUMBER(X,Y,0.07,FNUM,0.0,-1)
280  CONTINUE
      X=XOFF+0.2
      Y=YOFF-1.55
      CALL SYMBOL(X,Y,0.28,IL(1),0.0,16)
      X=X+16*0.28*6.0/7.0
      FSCALE=1.0*ISCALE
      CALL NUMBER(X,Y,0.28,FSCALE,0.0,-1)
      AN=1.0*N
      Y=Y-0.5
      X=XOFF+1.0
      CALL NUMBER(XOFF,Y,0.21,AN,0.0,-1)
      CALL SYMBOL(X,Y,0.21,IS(1),0.0,24)
      X=XOFF+2.2
      Y=YOFF-0.75
      CALL SYMBOL(X,Y,.07,ISX(1),0.0,12)
      X=XOFF+6.25
      Y=YOFF+3.2
      CALL SYMBOL(X,Y,.07,ISY(1),90.0,16)
      DO 320 K=1,N

```

```

X=XOFF+FACTOR*(EAST(K)-XMIN)
Y=YOFF+FACTOR*(NORTH(K)-YMIN)
CALL PLOT(X,Y,3)
CALL MARKER(2)
320 CONTINUE
WRITE(5,330) TITOLD,N,FNAME
WRITE(6,330) TITOLD,N,FNAME
330 FORMAT(1X,A8,5X,I4,' STATION LOCATIONS',8X,
1' PLOTFILE NAME ',A12)
CALL RSTR(2)
C*****
C IF END OF INPUT FILE THEN TERMINATE, ELSE GO ROUND AGAIN
C FOR NEXT GROUP
C*****
IF(END) GO TO 900
KOUNT=KOUNT+1
GO TO 20
600 END=.TRUE.
GO TO 200
900 CONTINUE
WRITE(6,910)
910 FORMAT(1X)
END

```



\*\*MID273

1 : 250000

162 STATION LOCATIONS, AMG

Figure 1.

